

//graphes : options courantes

```
A = gca(); // accès aux propriétés des axes
A.x_location="origin";A.y_location="origin"; // les axes se coupent en 0

// on ajoute un titre :
titre = input("donner le titre du graphe (caractères ASCII) : ","s");
A.title.text=titre;
A.title.font_size=3;
titre_ok = input("texte du titre correct (o/n) ? ","s");
while titre_ok == "n"
    titre = input("donner le titre du graphe : ","s");
    A.title.text=titre;
    titre_ok = input("titre correct (o/n) ? ","s");
end
postitre_ok = input("position correcte (o/n) ? ","s");
while postitre_ok == "n"
    x_titre = input("donner la position horizontale du titre : ");
    y_titre = input("donner la position verticale du titre : ");
    A.title.position=[x_titre,y_titre];
    postitre_ok = input("position correcte (o/n/a) [a = abandon, retour en position automatique] ? ","s");
end
if postitre_ok == "a"
A.title.auto_position = "on"
end

// on étiquette les axes :
abcsisse = input("étiquette des abscisses (caractères ASCII) : ","s");
ordonnee = input("étiquette des ordonnées (caractères ASCII) : ","s");
A.x_label.text=abcsisse;
A.y_label.text=ordonnee;
A.x_label.font_size=3;A.y_label.font_size=3;

// autres options : trait, fond, boîte, grille
A.children.children.thickness=2; //épaisseur courbe(s)
A.background=color(249,248,230);
A.box="on";
A.thickness=1;A.foreground=color(90,90,90);
A.grid=[16,16];
A.grid_style=[7,7];
A.grid_thickness=[1,1];
```